行列の簡約、固有値と固有ベクトル

1 URL Maxima マニュアル

多くのマニュアルが掲載されているので、参考にしてみると、より効果が高まるでしょう。

REF:https://flex.phys.tohoku.ac.jp/texi/maxima/maxima_26.html

マクロコマンド（eigen, vect）に含まれる関数：

```
load(eigen)
    cross(u,v) ;  \% outerproductinprod
    (u,v) ;  \%innerproduct
    gramschmidt(M),
    columnvector([aa,bb,cc]);
    eigenvalues(M), eivals(M);eigenvectors(M), eivects(M);
      [vals, vecs]: eigenvectors(M);
       for  i thru length(vals[1]) do
       disp( val[i] = vals[1][i], mult[i] = vals[2][i], vec[i] = vecs[1][i] );
    unitvector(v), uvect(v); vectorsimp(expr); similaritytransform(M);
load(vect)
    u \ensuremath{\sim} v  \% cross productexpress(expr),
    vectorsimp, scalefactors,potential, vectorlpotential
    charapoly(M,x);  \% determinant(M - diagmatrix(length(M), x));
```

固有値と固有ベクトルの計算は、線形代数の学ぶとき、重要な項目です。

2 ２つの行列を比較すると

与えられた行列を変形すると、本質的な不変量が明瞭になることをつぎの 2 つの行列について計算してみましょう。まず行列式の値が変わらないことに驚きませんか？ 2 つの行列 A, B を眺めてみましょう。P は仲介の役割をします。

(% i1) A:matrix([-13,-8,-4], [12, 7,4], [24, 16, 7]);

$$\begin{pmatrix} -13 & -8 & -4 \\ 12 & 7 & 4 \\ 24 & 16 & 7 \end{pmatrix} \quad (\%\ o1)$$

(% i2) P:matrix([1,1,2], [-2, -1, -3], [1,-2,0]);

$$\begin{pmatrix} 1 & 1 & 2 \\ -2 & -1 & -3 \\ 1 & -2 & 0 \end{pmatrix} \quad (\%\ o2)$$

(% i3) B:matrix([-1,0,0], [0,3,0],[0,0,-1]);

$$\begin{pmatrix} -1 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & -1 \end{pmatrix} \quad (\%\ o3)$$

(% i4) PV:invert(P);

$$\begin{pmatrix} -6 & -4 & -1 \\ -3 & -2 & -1 \\ 5 & 3 & 1 \end{pmatrix} \quad (\%\ o4)$$

(%i5) PV.A.P;

$$\begin{pmatrix} -1 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & -1 \end{pmatrix} \quad (\%o5)$$

(%i6) determinant(A);

$$3 \quad (\%o6)$$

(%i7) determinant(B);

$$3 \quad (\%o7)$$

(%i8) determinant(PV.A.P);

$$3 \quad (\%o8)$$

見かけはかなりの違いがあっても、本質的な行列の性質が同じなのです。この解析をすることが、行列の固有値とか固有ベクトルを計算することで明確になるのです。そのために固有多項式（charpoly）から計算してみましょう。この計算式は、行列式に変数 x をもちいると det |A - x I | を展開した多項式で、これを方程式とした解を求めます。

(%i11) charpoly(A,x);

$$\left((7-x)^2-64\right)(-x-13) \\ +8(12(7-x)-96) \\ -4(192-24(7-x)) \quad (\%o11)$$

展開して因数分解します。計算はたいへんですが、パソコンを使います。理論もしっかりと理解し、さらに手計算（鉛筆と紙）で確認をしておいてください。

(%i12) expand(%);

$$-x^3+x^2+5x+3 \quad (\%o12)$$

(%i13) factor(%);

$$-(x-3)(x+1)^2 \quad (\%o13)$$

このように $x=3$ とか $x=-1$ を得ています。この解は、固有値とよばれます。対角化した行列の対角成分にもなっています。重複しますが、最初から計算していきます。

3　内積と外積

ここでは内積 (inner product, scaler product), 外積（outer product, vector product, cross product）の計算をしておく。外積を cross とよぶ理由は、小行列の積を要素とするからである。ベクトル解析や電磁気学では、重要な事項です。

(%i7) load(eigen)$

(%i8) u : [a,b,c];

$$[a,b,c] \quad (\%o8)$$

(%i9) v: [x,y,z];

$$[x,y,z] \quad (\%o9)$$

“inprod” として命令を定めています。またドット積としても同じ結果となることを確認しています。

(%i10) inprod(u,v);

$$cz+by+ax \quad (\%o10)$$

(%i11) u.v;

$$cz+by+ax \quad (\%o11)$$

このような内積は直接に、u.v := sum(u[i]*v[i], i, 1, length(a)); % dot product と定義することもできます。定義式にはピリオド、コロン、イコールの記号を用います。外積、クロス積も同様に定められます。

(% i12) cross(u,v) := [u[2]*v[3] - u[3]*v[2],
u[3]*v[1] - u[1]*v[3],
u[1]*v[2] - u[2]*v[1]];

$$\mathrm{cross}(u,v) := [u_2 v_3 - u_3 v_2, u_3 v_1 - u_1 v_3, u_1 v_2 - u_2 v_1] \quad (\% \text{ o12})$$

(% i13) u[1];

$$a \quad (\% \text{ o13})$$

(% i14) v[3];

$$z \quad (\% \text{ o14})$$

マクロ “load(vect)” をすると、次のようにベクトルとして出力します。

(% i15) load(vect)$
(% i17) u~ v$
express(%);

$$[bz - cy, cx - az, ay - bx] \quad (\% \text{ o17})$$

“express” は表示という意味です
ベクトルのノルム（大きさ、長さ）は、norm 命令です。

(% i18) v: [a,b,c];

$$[a,b,c] \quad (\% \text{ o18})$$

(% i19) norm(v):= sqrt(sum(v[i]^2, i, 1, length(v)));

$$\mathrm{norm}(v) := \sqrt{\sum_{i=1}^{\mathrm{length}(v)} v_i^2} \quad (\% \text{ o19})$$

(% i20) norm(v);

$$\sqrt{c^2 + b^2 + a^2} \quad (\% \text{ o20})$$

4 グラムシュミットの直交化

グラムシュミットの直交化（Gram Schmidt orthogonalization）はベクトルの組 $\{\vec{v}_1, \vec{v}_2, \cdots\}$ から、互いに直交するベクトルを構成する方法です。与えられたベクトルが 1 次独立であれば、これらから構成しますが、考える次元に不足する一次独立なベクトルの個数であれば、追加しなければなりません。あらかじめ一次独立かどうか、その個数を調べることがある場合があります。ベクトルの組から作った行列の階数 (rank) などで調べます。3 次元空間において、3 個のベクトルの組に対しても、 3 個のベクトルが同一の平面に含まれるならば、3 番目のベクトルはこの平面の外に選ばなければ、 3 つのベクトル組で空間は表すことができません。任意のベクトルを表すベクトルの組は空間の基底（base）とよび、互いに直交するものが直交ベクトル、ベクトルの大きさ（長さ）が 1 となるように調整したものが正規 (normal) といい、これらが、正規直交基底系 (normal orthogonal base system) です。system とは組の概念です。

多くの場合、マクロコ命令として load(eigen)$ を入力して実行します。

例.1

(% i21) mX:
matrix([1,2,3], [9, 18, 30], [12, 48, 60]);

$$\begin{pmatrix} 1 & 2 & 3 \\ 9 & 18 & 30 \\ 12 & 48 & 60 \end{pmatrix} \quad (\% \text{ o21})$$

(% i22) vY: gramschmidt(mX);

$$\left[[1,2,3], \left[-\frac{3^2}{2\cdot 7}, -\frac{3^2}{7}, \frac{3\cdot 5}{2\cdot 7}\right], \left[-\frac{2^4 3}{5}, \frac{2^3 3}{5}, 0\right]\right] \quad (\% \text{ o22})$$

この出力結果では　計算の簡略化を因数に分解した形ですから、積の記号をわざと追加しています。注意してください。つぎのように　ev　にすれば計算し

てくれます。

(% i23) ev(vY);

$$\left[[1,2,3],\left[-\frac{9}{14},-\frac{9}{7},\frac{15}{14}\right],\left[-\frac{48}{5},\frac{24}{5},0\right]\right] \quad (\% \text{o23})$$

直交化の確認です。ベクトルの内積を計算すると、ゼロになります。

(% i24) vY[1];

$$[1,2,3] \quad (\% \text{o24})$$

(% i25) vY[2];

$$\left[-\frac{3^2}{2\cdot 7},-\frac{3^2}{7},\frac{3\cdot 5}{2\cdot 7}\right] \quad (\% \text{o25})$$

(% i26) vY[1].vY[2];

$$-\frac{3^2}{2\cdot 7}-\frac{23^2}{7}+\frac{45}{14} \quad (\% \text{o26})$$

(% i27) ev(%);

$$0 \quad (\% \text{o27})$$

例 2

(% i28) mX2:matrix([1,1,0], [2, 0, 1], [0, 2, 1]);

$$\begin{pmatrix} 1 & 1 & 0 \\ 2 & 0 & 1 \\ 0 & 2 & 1 \end{pmatrix} \quad (\% \text{o28})$$

(% i29) vY2 : gramschmidt(mX2);

$$\left[[1,1,0],[1,-1,1],\left[-\frac{2}{3},\frac{2}{3},\frac{2^2}{3}\right]\right] \quad (\% \text{o29})$$

(% i30) vY21:vY2[1];

$$[1,1,0] \quad (\% \text{o30})$$

(% i31) vY22:vY2[2];

$$[1,-1,1] \quad (\% \text{o31})$$

(% i32) nY21:sqrt(sum(vY2[1][i]^2, i, 1, length(vY2[1])));

$$\sqrt{2} \quad (\% \text{o32})$$

(% i33) gY21: vY22 / nY21;

$$\left[\frac{1}{\sqrt{2}},-\frac{1}{\sqrt{2}},\frac{1}{\sqrt{2}}\right] \quad (\% \text{o33})$$

ベクトルのノルム（大きさ、長さ）を正規化 normalize（ノルムが 1）となるようにします。また内積もゼロとなり、直交しています。

(% i34) nY22:norm(vY22);

$$\sqrt{3} \quad (\% \text{o34})$$

(% i35) vY21.vY22;

$$0 \quad (\% \text{o35})$$

5 固有値と固有ベクトルの計算

例題 1 多重度 (multi = 1) で重複解でない場合です。変数 vals と vecs を定義して出力を見やすくする工夫をしています。

(% i36) M1:matrix([11,-1],[1,7]);

$$\begin{pmatrix} 11 & -1 \\ 1 & 7 \end{pmatrix} \quad (\% \text{o36})$$

(% i37) [vals,vecs]:eigenvectors(M1);

$$\left[\left[\left[9-\sqrt{3},\sqrt{3}+9\right],[1,1]\right],\left[\left[\left[1,\sqrt{3}+2\right]\right],\left[\left[1,2-\sqrt{3}\right]\right]\right]\right] \quad (\% \text{o37})$$

(% i38) *for i thru length (vals[1]) do disp(val[i] = vals[1][i], mult[i] = vals[2][i], vec[i] = vecs[i]);*

$$\mathrm{val}_1 = 9-\sqrt{3}$$
$$\mathrm{mult}_1 = 1$$
$$\mathrm{vec}_1 = \left[\left[1,\sqrt{3}+2\right]\right]$$
$$\mathrm{val}_2 = \sqrt{3}+9$$
$$\mathrm{mult}_2 = 1$$
$$\mathrm{vec}_2 = \left[\left[1,2-\sqrt{3}\right]\right]$$

done (% o38)

例題 2 多重度 (multi) をもつ場合です

(% i39) M2:matrix([0,1,0,0], [0,0,0,0], [0,0,2,0], [0,0,0,2]);

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & 0 & 2 \end{pmatrix} \qquad (\%\ o39)$$

(% i40) [vals,vecs]:eigenvectors(M2);

$$[[[0,2],[2,2]],[[[1,0,0,0]],[[0,0,1,0],[0,0,0,1]]]] \qquad (\%\ o40)$$

(% i41) for i thru length (vals[1]) do disp(val[i] = vals[1][i], mult[i]= vals[2][i], vec[i]=vecs[i]);

$$\mathrm{val}_1 = 0$$
$$\mathrm{mult}_1 = 2$$
$$\mathrm{vec}_1 = [[1,0,0,0]]$$
$$\mathrm{val}_2 = 2$$
$$\mathrm{mult}_2 = 2$$
$$\mathrm{vec}_2 = [[0,0,1,0],[0,0,0,1]]$$

done (% o41)

例題 3 テキスト p.178 例題 1

(% i1) A:matrix([3,2,4], [2,0,2], [4,2,3]);

$$\begin{pmatrix} 3 & 2 & 4 \\ 2 & 0 & 2 \\ 4 & 2 & 3 \end{pmatrix} \qquad (\%\ o1)$$

(% i2) charapoly(A,x); /* spell miss :*/

$$\mathrm{charapoly}\left(\begin{pmatrix} 3 & 2 & 4 \\ 2 & 0 & 2 \\ 4 & 2 & 3 \end{pmatrix}, x\right) \qquad (\%\ o2)$$

もしスペルミスをして、プログラムが理解できない場合の結果を表しています。そのままの単語が出ています。

(% i3) load(eigen)$

(% i4) charpoly(A,x);

$$(3-x)\,(-(3-x)\,x-4)+4\,(4x+4)-2\,(2\,(3-x)-8) \qquad (\%\ o4)$$

(% i5) cP:expand(charpoly(A,x));

$$-x^3+6x^2+15x+8 \qquad (\%\ o5)$$

(% i6) factor(%);

$$-(x-8)\,(x+1)^2 \qquad (\%\ o6)$$

(% i7) solve((8*iden(3) - A).[x,y,z]= [0,0,0], [x,y,z]);

$$[] \qquad (\%\ o7)$$

これも入力ミスですね.

(% i8) [vals,vecs]:eigenvectors(A);

$$\left[[[8,-1],[1,2]],\left[\left[\left[1,\frac{1}{2},1\right]\right],\left[[1,0,-1],\left[0,1,-\frac{1}{2}\right]\right]\right]\right] \qquad (\%\ o8)$$

(% i9) for i thru length (vals[1]) do disp(val[i] = vals[1][i], mult[i]= vals[2][i], vec[i]=vecs[i]);

$val_1 = 8$
$mult_1 = 1$
$vec_1 = \left[\left[1, \frac{1}{2}, 1\right]\right] val_2 = -1$
$mult_2 = 2$
$vec_2 = \left[[1, 0, -1], \left[0, 1, -\frac{1}{2}\right]\right]$

done (% o9)

/* confirmation */

$tr(matrixA) = 3 + 0 + 0 = 3,$
$vec_1 * mult_1 + vec_2 * mult_2 = 8 * 1 + (-1) * 2 = 6$
$det(marixA) = -8,$
$vec_1^{mult_1} * vec_2^{mult_2} = 8 * (-1) * (-1)^2 = -8$

(% i10) determinant(A);

8 (% o10)

例題 5 章 A 1 (6) 179 ページ

(% i1) load(eigen)$

(% o1)

(% i2) A: matrix([1,0,0,0],[0,0,1,-1],[0,-2,3,-1], [0,-1,1,1]);

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & -1 \\ 0 & -2 & 3 & -1 \\ 0 & -1 & 1 & 1 \end{pmatrix} \quad (\%\ o2)$$

(% i3) cP:charpoly(A,x);

$(1-x)(-((1-x)(3-x)+1)x + x + 2(1-x))$ (% o3)

(% i4) expand(%);

$x^4 - 5x^3 + 9x^2 - 7x + 2$ (% o4)

(% i5) factor(%);

$(x-2)(x-1)^3$ (% o5)

(% i6) [vals,vecs]:eigenvectors(A);

$[[[2, 1], [1, 3]], [[[0, 0, 1, 1]], [[1, 0, 0, 0], [0, 1, 1, 0]]]]$ (% o6)

(% i7) for i thru length (vals[1]) do disp(val[i] = vals[1][i], mult[i]= vals[2][i], vec[i]=vecs[i]);

$val_1 = 2$
$mult_1 = 1$
$vec_1 = [[0, 0, 1, 1]]$
$val_2 = 1$
$mult_2 = 3$
$vec_2 = [[1, 0, 0, 0], [0, 1, 1, 0]]$

done (% o7)

(% i8) determinant(A);

2 (% o8)

/* 2*1*1*1 = 2; product of eigen values = determinant */

(% i9) 1+0+3+1;

5 (% o9)

/* 2 + 1+1+1 = 5 ; sum of eigen value = trace */

6 固有値と固有ベクトル（２）

正方行列 (square matrix) A に対して、ベクトル $\vec{x}(\neq \vec{0})$ と定数 λ を

$$A\vec{x} = \lambda\vec{x}$$

となるよう定められるとき、スカラー値 λ を **固有値** (eigen value), ベクトル $\vec{x}$ を **固有ベクトル** (eigen vector) という。固有値は複数個が存在し、各々の値に対して、ベクトルが対応する。

n 次正方行列 $A = (a_{ij})$ とし、ベクトル $\vec{x}$ とすると

$$A\vec{x} = \lambda\vec{x} \quad \Leftrightarrow \quad (A - \lambda I)\vec{x} = \vec{0} \quad \Leftrightarrow \quad \det|A - \lambda I| = 0\,(\vec{x} \neq \vec{0})$$

右辺の関係は連立方程式であり、行列式 (determinant) は展開することで、変数 λ の n 次多項式となる。これを**固有多項式** (characteristic polynomial) とよび、因数分解して、n 個の解を $\lambda_1, \lambda_2, \cdots \lambda_n$ と表す。

$$\varphi_A(x) = (x - \lambda_1)(x - \lambda_2)\cdots(x - \lambda_n) = (x - \lambda_1)^{m_1}(x - \lambda_2)^{m_2}\cdots(x - \lambda_r)^{m_r} \quad (r \leq n)$$

ここで、重複解を考慮して、$\{m_i, i = 1, 2, \cdots\}$ を固有値 λ_i の**重複度** (multiplicity) という。

例１. テキスト 174 ページ

(% i11) load(eigen)$

(% i12) mA:matrix([4,3],[1,2]);

$$\begin{pmatrix} 4 & 3 \\ 1 & 2 \end{pmatrix} \qquad (\%\ o12)$$

(% i13) charpoly(mA,x);

$$(2 - x)\,(4 - x) - 3 \qquad (\%\ o13)$$

(% i14) expand(%);

$$x^2 - 6x + 5 \qquad (\%\ o14)$$

(% i15) factor(%);

$$(x - 5)\,(x - 1) \qquad (\%\ o15)$$

(% i16) [vals,vecs] : eigenvectors(mA);

$$\left[[[1, 5], [1, 1]], \left[[[1, -1]], \left[\left[1, \frac{1}{3}\right]\right]\right]\right] \qquad (\%\ o16)$$

(% i17) for i thru length(vals[1]) do disp(val[i] = vals[1][i], mult[i] = vals[2][i], vec[i] = vecs[i]);

$$\mathrm{val}_1 = 1$$
$$\mathrm{mult}_1 = 1$$
$$\mathrm{vec}_1 = [[1, -1]]$$
$$\mathrm{val}_2 = 5$$
$$\mathrm{mult}_2 = 1$$
$$\mathrm{vec}_2 = \left[\left[1, \tfrac{1}{3}\right]\right]$$

$$\mathrm{done} \qquad (\%\ o17)$$

例題１ テキスト　１７６ページ

(% i3) load(eigen)$

(% i4) mA:matrix([3,2,4],[2,0,2],[4,2,3]);

$$\begin{pmatrix} 3 & 2 & 4 \\ 2 & 0 & 2 \\ 4 & 2 & 3 \end{pmatrix} \qquad (\%\ o4)$$

(% i5) charpoly(mA,x);

$$(3 - x)\,(-(3 - x)\,x - 4) + 4\,(4x + 4) - 2\,(2\,(3 - x) - 8) \qquad (\%\ o5)$$

(% i6) expand(%);

$$-x^3 + 6x^2 + 15x + 8 \qquad (\%\ o6)$$

(% i7) factor(%);

$$-(x - 8)\,(x + 1)^2 \qquad (\%\ o7)$$

(% i8) [vals,vecs] : eigenvectors(mA);

$$\left[[[8, -1], [1, 2]], \left[\left[\left[1, \frac{1}{2}, 1\right]\right], \left[[1, 0, -1], \left[0, 1, -\frac{1}{2}\right]\right]\right]\right] \qquad (\%\ o8)$$

(% i9) for i thru length(vals[1]) do disp(
val[i] = vals[1][i],
mult[i] = vals[2][i], vec[i] = vecs[i]);

$$\mathrm{val}_1 = 8$$
$$\mathrm{mult}_1 = 1$$
$$\mathrm{vec}_1 = \left[\left[1, \tfrac{1}{2}, 1\right]\right]$$
$$\mathrm{val}_2 = -1$$
$$\mathrm{mult}_2 = 2$$
$$\mathrm{vec}_2 = \left[[1, 0, -1], \left[0, 1, -\tfrac{1}{2}\right]\right]$$

$$\mathrm{done} \qquad (\%\ o9)$$

この例題では固有値 $vals_2 = -1$ の重複度 $mult_2 =$

2 であり、２個のベクトルによって張られる空間となる（解ベクトル）で、$\left\langle \begin{pmatrix} 1 \\ 0 \\ -1 \end{pmatrix}, \begin{pmatrix} 0 \\ 2 \\ -1 \end{pmatrix} \right\rangle$ を表している。これはテキストでは $\left\langle \begin{pmatrix} 1 \\ -2 \\ 0 \end{pmatrix}, \begin{pmatrix} 0 \\ -2 \\ 1 \end{pmatrix} \right\rangle$ となっているが、同じであることを確かめてみよ。

(ヒント)；第２成分と第３成分を比較すると同じであることは明白であろう。ベクトルの長さと変えたり、軸を回転する（軸の順序を取り替える）ことをしても同じ空間である。